<?xml version="1.0" encoding="utf-8" ?>
<xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform" version="1.0">
<xsl:variable name="page"><xsl:value-of select="dataset/@p"/></xsl:variable>
<xsl:variable name="size"><xsl:value-of select="dataset/@s"/></xsl:variable>
<xsl:template match="/dataset/table[1]">
<xsl:choose>
	<xsl:when test="count(row)>0">
		<xsl:for-each select="row">
			<p>
				<input type="checkbox" name="message" value="{id}" />
				<layer><xsl:value-of select="posttime" /></layer>
				<span>
					<xsl:call-template name="captureUBB">
						<xsl:with-param name="processText">
							<xsl:value-of select="content"/>
						</xsl:with-param>
					</xsl:call-template>
				</span>
			</p>
		</xsl:for-each>
	</xsl:when>
	<xsl:otherwise>
		<div class="manager_message_null">
			当前没有消息
		</div>
	</xsl:otherwise>
</xsl:choose>
</xsl:template>
<xsl:template match="/dataset/table[2]/row">
	<xsl:if test="number(count)>number($size)">
		<div class="syblog_pager">
			<div class="syblog_pager_right_bar">
				<div class="syblog_pager_header_bar"><span class="left"></span><span class="right"></span></div>
					<div class="syblog_pager_content">
						<xsl:call-template name="pager">
							<xsl:with-param name="p">
								<xsl:value-of select="ceiling(number(count) div number($size))"/>
							</xsl:with-param>
						</xsl:call-template>
					</div>
				<div class="syblog_pager_footer_bar footer_bar"><span class="left"></span><span class="right"></span></div>
			</div>
		</div>
	</xsl:if>
</xsl:template>
<xsl:template name="pager">
	<xsl:param name="p"/>
	<xsl:if test="$p>1">
		<xsl:if test="($page)>1">
			<a href="#" onclick="Manager.getSysMessage({number($page)-1});return false;">上一页</a>
		</xsl:if>
		<xsl:call-template name="loop">
			<xsl:with-param name="min">
				<xsl:choose>
					<xsl:when test="$p>10 and $page>5">
						<xsl:choose>
							<xsl:when test="$p>$page+5">
								<xsl:value-of select="number($page)-5"/>
							</xsl:when>
							<xsl:otherwise><xsl:value-of select="number($p)-10"/></xsl:otherwise>
						</xsl:choose>
					</xsl:when>
					<xsl:otherwise>1</xsl:otherwise>
				</xsl:choose>
			</xsl:with-param>
			<xsl:with-param name="max">
				<xsl:choose>
					<xsl:when test="$p>10">
						<xsl:choose>
							<xsl:when test="$p>$page+5">
								<xsl:choose>
									<xsl:when test="$page>5">
										<xsl:value-of select="number($page)+5"/>
									</xsl:when>
									<xsl:otherwise>10</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:otherwise><xsl:value-of select="$p"/></xsl:otherwise>
						</xsl:choose>
					</xsl:when>
					<xsl:otherwise><xsl:value-of select="$p"/></xsl:otherwise>
				</xsl:choose>
			</xsl:with-param>
		</xsl:call-template>
		<xsl:if test="$p>number($page)">
			<a href="#" onclick="Manager.getSysMessage({number($page)+1});return false;">下一页</a>
		</xsl:if>
	</xsl:if>
</xsl:template>
<xsl:template name="loop">
	<xsl:param name="min"/>
	<xsl:param name="max"/>
	<xsl:if test="$max>=$min">
		<xsl:choose>
			<xsl:when test="number($min)=number($page)">
				<xsl:value-of select="$min"/>
			</xsl:when>
			<xsl:otherwise>
				<a href="#" onclick="Manager.getSysMessage({$min});return false;">[<xsl:value-of select="$min" />]</a>
			</xsl:otherwise>
		</xsl:choose>
		<xsl:call-template name="loop">
			<xsl:with-param name="min">
				<xsl:value-of select="number($min)+1"/>
			</xsl:with-param>
			<xsl:with-param name="max">
				<xsl:value-of select="$max"/>
			</xsl:with-param>
		</xsl:call-template>
	</xsl:if>
</xsl:template>

  <xsl:template name="captureUBB">
    <xsl:param name="processText"/>
    <xsl:variable name="remaining_after_end_ubb" select="substring-after($processText, '[/')"/>
    <xsl:variable name="remainingContent_after_ubb" select="substring-after($remaining_after_end_ubb, ']')"/>

    <xsl:choose>
      <xsl:when test="contains($processText, '[') and contains($processText, ']')">
        <xsl:choose>
          <xsl:when test="starts-with($processText, '[')">
            <xsl:call-template name="processUBB">
              <xsl:with-param name="ubbType" select="translate(substring-after(substring-before($processText, ']'), '['),'ABCDEFGHIJKLMNOPQRSTUVWXYZ','abcdefghijklmnopqrstuvwxyz')"/>
              <xsl:with-param name="content" select="substring-after(substring-before($processText, '[/'), ']')"/>
              <xsl:with-param name="backingContent" select="substring-after($processText, ']')"/>
              <xsl:with-param name="remainingContent" select="$remainingContent_after_ubb"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:otherwise>
            <xsl:value-of select="substring-before($processText, '[')"/>
            <xsl:call-template name="captureUBB">
              <xsl:with-param name="processText" select="concat('[',substring-after($processText, '['))"/>
            </xsl:call-template>
          </xsl:otherwise>
        </xsl:choose>
      </xsl:when>
      <xsl:otherwise>
        <xsl:value-of select="$processText"/>
      </xsl:otherwise>
    </xsl:choose>
  </xsl:template>

  <xsl:template name="processUBB">
    <xsl:param name="ubbType"/>
    <xsl:param name="content"/>
    <xsl:param name="backingContent"/>
    <xsl:param name="remainingContent"/>

    <xsl:choose>
      <xsl:when test="$ubbType=translate(substring-before(substring-after($backingContent, '[/'),']'),'ABCDEFGHIJKLMNOPQRSTUVWXYZ','abcdefghijklmnopqrstuvwxyz')">
        <xsl:choose>
          <xsl:when test="$ubbType = 'b'">
            <xsl:call-template name="b">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'email'">
            <xsl:call-template name="email">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'fly'">
            <xsl:call-template name="fly">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'i'">
            <xsl:call-template name="i">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'img'">
            <xsl:call-template name="img">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'move'">
            <xsl:call-template name="move">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'quote'">
            <xsl:call-template name="quote">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'u'">
            <xsl:call-template name="u">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:when test="$ubbType = 'url'">
            <xsl:call-template name="url">
              <xsl:with-param name="content" select="$content"/>
              <xsl:with-param name="remainingContent" select="$remainingContent"/>
            </xsl:call-template>
          </xsl:when>
          <xsl:otherwise>
            <xsl:text>[</xsl:text>
            <xsl:value-of select="$ubbType"/>
            <xsl:text>]</xsl:text>
            <xsl:call-template name="captureUBB">
              <xsl:with-param name="processText" select="$backingContent"/>
            </xsl:call-template>
          </xsl:otherwise>
        </xsl:choose>
      </xsl:when>
      <xsl:when test="contains($ubbType, '=') and substring-before($ubbType, '=')=translate(substring-before(substring-after($backingContent, '[/'),']'),'ABCDEFGHIJKLMNOPQRSTUVWXYZ','abcdefghijklmnopqrstuvwxyz')">
        <xsl:call-template name="processUBB2">
          <xsl:with-param name="ubbType" select="substring-before($ubbType, '=')"/>
          <xsl:with-param name="ubbParam" select="substring-after($ubbType, '=')"/>
          <xsl:with-param name="content" select="$content"/>
          <xsl:with-param name="backingContent" select="$backingContent"/>
          <xsl:with-param name="remainingContent" select="$remainingContent"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:otherwise>
        <xsl:text>[</xsl:text>
        <xsl:value-of select="$ubbType"/>
        <xsl:text>]</xsl:text>
        <xsl:call-template name="captureUBB">
          <xsl:with-param name="processText" select="$backingContent"/>
        </xsl:call-template>
      </xsl:otherwise>
    </xsl:choose>
  </xsl:template>

  <xsl:template name="processUBB2">
    <xsl:param name="ubbType"/>
    <xsl:param name="ubbParam"/>
    <xsl:param name="content"/>
    <xsl:param name="backingContent"/>
    <xsl:param name="remainingContent"/>

    <xsl:choose>
      <xsl:when test="$ubbType = 'align'">
        <xsl:call-template name="align">
          <xsl:with-param name="param" select="$ubbParam"/>
          <xsl:with-param name="content" select="$content"/>
          <xsl:with-param name="remainingContent" select="$remainingContent"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:when test="$ubbType = 'color'">
        <xsl:call-template name="color">
          <xsl:with-param name="param" select="$ubbParam"/>
          <xsl:with-param name="content" select="$content"/>
          <xsl:with-param name="remainingContent" select="$remainingContent"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:when test="$ubbType = 'email'">
        <xsl:call-template name="email2">
          <xsl:with-param name="param" select="$ubbParam"/>
          <xsl:with-param name="content" select="$content"/>
          <xsl:with-param name="remainingContent" select="$remainingContent"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:when test="$ubbType = 'face'">
        <xsl:call-template name="face">
          <xsl:with-param name="param" select="$ubbParam"/>
          <xsl:with-param name="content" select="$content"/>
          <xsl:with-param name="remainingContent" select="$remainingContent"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:when test="$ubbType = 'size'">
        <xsl:call-template name="size">
          <xsl:with-param name="param" select="$ubbParam"/>
          <xsl:with-param name="content" select="$content"/>
          <xsl:with-param name="remainingContent" select="$remainingContent"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:when test="$ubbType = 'url'">
        <xsl:call-template name="url2">
          <xsl:with-param name="param" select="$ubbParam"/>
          <xsl:with-param name="content" select="$content"/>
          <xsl:with-param name="remainingContent" select="$remainingContent"/>
        </xsl:call-template>
      </xsl:when>
      <xsl:otherwise>
        <xsl:text>[</xsl:text>
        <xsl:value-of select="$ubbType"/>
        <xsl:text>=</xsl:text>
        <xsl:value-of select="$ubbParam"/>
        <xsl:text>]</xsl:text>
        <xsl:call-template name="captureUBB">
          <xsl:with-param name="processText" select="$backingContent"/>
        </xsl:call-template>
      </xsl:otherwise>
    </xsl:choose>
  </xsl:template>

  <xsl:template name="b">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <b>
      <xsl:value-of select="$content"/>
    </b>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="email">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <a href="mailto:{$content}">
      <xsl:value-of select="$content"/>
    </a>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="fly">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <marquee behavior="alternate">
      <xsl:value-of select="$content"/>
    </marquee>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="i">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <i>
      <xsl:value-of select="$content"/>
    </i>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="img">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <img _src="{$content}"/>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="move">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <marquee>
      <xsl:value-of select="$content"/>
    </marquee>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="quote">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <blockquote>
      <i>
        <xsl:value-of select="$content"/>
      </i>
    </blockquote>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="u">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <u>
      <xsl:value-of select="$content"/>
    </u>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="url">
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <a href="{$content}" target="_blank">
      <xsl:value-of select="$content"/>
    </a>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="align">
    <xsl:param name="param"/>
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <p align="{$param}">
      <xsl:value-of select="$content"/>
    </p>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="color">
    <xsl:param name="param"/>
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <font color="{$param}">
      <xsl:value-of select="$content"/>
    </font>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="email2">
    <xsl:param name="param"/>
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <a href="mailto:{$param}">
      <xsl:value-of select="$content"/>
    </a>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="face">
    <xsl:param name="param"/>
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <font face="{$param}">
      <xsl:value-of select="$content"/>
    </font>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="size">
    <xsl:param name="param"/>
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <font size="{$param}">
      <xsl:value-of select="$content"/>
    </font>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>

  <xsl:template name="url2">
    <xsl:param name="param"/>
    <xsl:param name="content"/>
    <xsl:param name="remainingContent"/>
    <a href="{$param}" target="_blank">
      <xsl:value-of select="$content"/>
    </a>
    <xsl:call-template name="captureUBB">
      <xsl:with-param name="processText" select="$remainingContent"/>
    </xsl:call-template>
  </xsl:template>
</xsl:stylesheet>